

intro 7 — A Rosetta stone for extended regression commands

[Description](#)[Remarks and examples](#)[Also see](#)

Description

If you already are familiar with some or all of Stata's other commands that fit models with endogenous covariates, sample selection, and treatment effects, this entry shows you how to use that knowledge to fit equivalent models using ERMs.

Remarks and examples

[stata.com](#)

Aside from providing a single coherent framework that allows complications to be combined, ERMs use similar syntax and the resulting models have the same interpretation.

In most cases, the estimation method used by the ERM commands and that used by other estimators to fit the same model produce results that are the same. Typically, there are small numerical differences because the optimization is different. Also, ancillary parameters, such as variances of errors, are sometimes parameterized differently. In some cases, a different estimation method is used. In this case, results will be asymptotically equivalent, but in finite samples, results will differ.

The table below provides a basic guide for the correspondence between Stata commands you may already be familiar with and the ERM commands.

Command you know	Equivalent extended regression command
Linear regression with endogenous covariate <code>ivregress liml y1 x (y2 = z)</code>	<code>eregress y1 x, endogenous(y2 = z x)</code>
Probit model with endogenous covariate <code>ivprobit y1 x (y2 = z)</code>	<code>eprobit y1 x, endogenous(y2 = z x)</code>
Tobit model with endogenous covariate <code>ivtobit y1 x (y2 = z), ll(0) ul(20)</code>	<code>generate y1_ll = y1</code> <code>replace y1_ll = . if y1 <= 0</code> <code>generate y1_ul = y1</code> <code>replace y1_ul = . if y1 >= 20 & y1 < .</code> <code>eintreg y1_ll y1_ul x, endogenous(y2 = z x)</code>
Linear regression with exogenous treatment <code>teffects ra (y x1 x2) (t1)</code>	<code>eregress y x1 x2, extreat(t1) vce(robust)</code> <code>estat teffects</code>
Probit model with exogenous treatment <code>teffects ra (y x1 x2, probit) (t1)</code>	<code>eprobit y x1 x2, extreat(t1) vce(robust)</code> <code>estat teffects</code>

Linear regression with endogenous treatment

```
eregress y x, treat(t1 = x z)                eregress y x, entreat(t1 = x z, nointeract)
```

Linear regression with sample selection

```
heckman y x, select(s1 = x z)                eregress y x, select(s1 = x z)
```

Probit model with sample selection

```
heckprobit y x, select(s1 = x z)             eprobit y x, select(s1 = x z)
```

Ordered probit model with sample selection

```
heckoprobit y x, select(s1 = x z)           eoprobit y x, select(s1 = x z)
```

You can build on the basic syntax of the ERM commands by combining options and suboptions, giving you the flexibility to fit a myriad of models. Here is a short list of what you might try.

Linear regression with a continuous endogenous covariate but where the exogenous variable is not included as an instrument

```
. eregress y1 x, endogenous(y2 = z1)
```

Linear regression with two continuous endogenous covariates

```
. eregress y1 x, endogenous(y2 y3 = z1 x)
```

As above, but with different instruments for different endogenous covariates

```
. eregress y1 x, endogenous(y2 = z1 x) endogenous(y3 = z2 x)
```

As above, but with one endogenous covariate being binary

```
. eregress y1 x, endogenous(y2 = z1 x) endogenous(y4 = z3 x, probit)
```

Linear regression with a continuous endogenous covariate and an endogenous treatment

```
. eregress y1 x, endogenous(y2 = z1 x) entreat(t1 = z3 x)
```

As above, but instead include a multivalued treatment

```
. eregress y1 x, endogenous(y2 = z1 x) entreat(t2 = z3 x, oprobit)
```

As above, and also allow for endogenous selection

```
. eregress y1 x, endogenous(y2 = z1 x) entreat(t2 = z3 x, oprobit) ///  
  select(s1 = w x)
```

As above, but where censoring of variable `s2` indicates selection status

```
. eregress y1 x, endogenous(y2 = z1 x) entreat(t2 = z3 x, oprobit) ///  
  tobitselect(s2 = w x)
```

`eprobit` or `eoprobit` may be directly substituted for `eregress` above to fit a probit or ordered probit regression when `y1` is binary or ordinal. To fit a tobit or interval regression, you must use `eintreg` and specify two dependent variables containing the upper and lower bounds of the interval in place of `y1`.

Also see

[ERM] [intro 1](#) — An introduction to the ERM commands

[ERM] [intro 8](#) — Conceptual introduction via worked example